

Jak na propojení Excelu s Outlookem a proč vlastně? Po článku [Jak na propojení Excelu s Wordem](#) je další na řadě poštovní aplikace. Na čtyřech příkladech si ukážeme, čím si tyto dvě aplikace z balíčku Microsoft Office mohou být užitečné.

Hned na začátku něco, co sám potřebuji velmi často – seznam e-mailových adres, které se potulují v mém Outlooku.

```

1 'Tools / References / Microsoft Outlook xx.x Object Library
2
3 Private Sub OutlookNacistAdresyZAdresare()
4
5     Dim objOutlook As Object
6     Dim objAddressList As Object
7     Dim objAddressEntry As Object
8
9     Dim i As Long
10    Dim lngPocetPolozek As Long
11
12    Dim arrAdresy()
13
14    Set objOutlook = CreateObject("Outlook.Application")
15    Set objAddressList = objOutlook.Session.AddressLists("Kontakty")
16
17    Application.ScreenUpdating = False
18
19    'počet položek v adresáři
20    lngPocetPolozek = objAddressList.AddressEntries.Count
21
22    'dimenzování pole
23    ReDim arrAdresy(1 To lngPocetPolozek, 1 To 2)
24
25    'prokaždý záznam v adresáři
26    For Each objAddressEntry In objAddressList.AddressEntries
27
28        'počítadlo
29        i = i + 1
30        'jméno
31        arrAdresy(i, 1) = objAddressEntry.Name
32        'adresa
33        arrAdresy(i, 2) = objAddressEntry.Address
34
35        'Microsoft Exchange
36        'Set objAddressEntryDetail = objAddressEntry.GetExchangeUser
37
38    Next objAddressEntry
39
40    'vložení adres do listu
41    wshAdresy.Cells(1).Resize(lngPocetPolozek, 2) = arrAdresy
42
43    Application.ScreenUpdating = True
44
45    Set objAddressList = Nothing
46    Set objOutlook = Nothing
47
48 End Sub

```

Pokud jsou vaše kontakty uloženy v rámci Exchange, detail kontaktů najdete pod objektem `objAddressEntryDetail` (`PropertyAccessor`, `GetProperty`, schema a [Microsoft Exchange Property Tags](#)). To není můj případ a tak není jednoduché se dostat k jiným informacím, než je jméno a adresa. Uvedená procedura má i další háček. Outlook není dostatečně aktivní a ani nemotivuje uživatele v tom, aby odesílatele přidávali do svých adresářů. Takže v momentě, kdy chcete projít vaše kontakty, je v adresáři jednoduše nemáte. Proto nabízím i verzi, která vytáhá e-maily z doručených zpráv.

```

1 'Tools / References / Microsoft Outlook xx.x Object Library
2
3 Sub OutlookNacistAdresyZDorucenychMailu()
4
5     Dim i As Long
6     Dim lngPocetPolozek As Long
7
8     Dim arrAdresy()
9
10    Dim objOutlook As Object
11    Dim objNameSpace As Object
12    Dim objFolder As Object
13    Dim objItem As Object
14
15    Set objOutlook = New Outlook.Application
16    Set objNameSpace = objOutlook.GetNamespace("MAPI")
17    'olFolderInbox ... 6
18    Set objFolder = objNameSpace.GetDefaultFolder(olFolderInbox)
19
20    'počet položek v adresáři
21    lngPocetPolozek = objFolder.Items.Count
22
23    'dimenzování pole
24    ReDim arrAdresy(1 To lngPocetPolozek, 1 To 1)
25
26    For Each objItem In objFolder.Items
27
28        'je položka typu e-mail?
29        'olMail ... 43
30        If objItem.Class = olMail Then
31            'počítadlo
32            i = i + 1
33            'přidání adresy odesílatele do pole
34            arrAdresy(i, 1) = objItem.SenderEmailAddress
35        End If
36
37    Next objItem
38
39    Application.ScreenUpdating = False
40
41    With wshAdresy
42
43        'vložení adres do listu
44        .Cells(1).Resize(lngPocetPolozek, 1) = arrAdresy
45
46        'odstranění duplicit
47        .Range("A:A").RemoveDuplicates Columns:=1, Header:=xlNo
48
49    End With
50
51    'setřídění listu A-Z
52    With wshAdresy.Sort
53        .SortFields.Clear
54        .SortFields.Add Key:=Range("A1"), SortOn:=xlSortOnValues, _
55            Order:=xlAscending, DataOption:=xlSortNormal
56        .SetRange Range("A:A")
57        .Header = xlGuess
58        .MatchCase = False
59        .Orientation = xlTopToBottom
60        .SortMethod = xlPinYin
61        .Apply
62    End With
63
64    Application.ScreenUpdating = True
65
66    'odstranění proměnných z paměti
67    Set objFolder = Nothing
68    Set objNameSpace = Nothing
69    Set objOutlook = Nothing
70
71 End Sub

```

Po převzetí adres je samozřejmě potřeba řešit duplicity a není na škodu si adresy abecedně seřadit. I tak vám ještě zůstane trocha ruční práce – spam, který prošel, „no reply“ adresy apod. A hodit se vám může i tip na zřetězení. Postačí prostý odkaz na oblast vzorcem, stisk F9 (dojde k nahrazení adresy oblasti za hodnoty) a zbavení se drobného balastu. Položky oddělené středníky lze rovnou aplikovat ve zprávě Outlooku. První adresu věnujte políčku Komu a ostatní umístěte do skryté kopie (tlačítko Kopie a v otevřeném dialogu políčko Skrytá). Hromadné e-maily prosím rozesílejte jen v rozumné míře...



Zřetězení e-mailových adres pro hromadnou korespondenci

Přítvrdíme. Co si třeba projít všechny zprávy a vypsát jen takové, které obsahují v předmětu slovo „Excel“?

```

1 'Tools / References / Microsoft Outlook xx.x Object Library
2
3
4 Sub OutlookNacistZpravyDlePredmetu()
5
6 Dim i As Long
7 Dim lngPocetPolozek As Long
8
9
10 Dim arrPoleData()
11
12 Dim objOutlook As Object
13 Dim objNameSpace As Object
14 Dim objFolder As Object
15 Dim objItem As Object
16
17 Set objOutlook = New Outlook.Application
18 Set objNameSpace = objOutlook.GetNamespace("MAPI")
19 'olFolderInbox ... 6
20 Set objFolder = objNameSpace.GetDefaultFolder(olFolderInbox)
21
22
23 'aktivace patřičného listu
24 wshZpravy.Activate
25
26
27 'počet položek v adresáři
28 lngPocetPolozek = objFolder.Items.Count
29
30
31 'dimenzování pole
32 ReDim arrPoleData(1 To lngPocetPolozek, 1 To 4)
33
34 For Each objItem In objFolder.Items
35
36 'je položka typu e-mail
37 'a obsahuje předmět slovo "Excel"?
38 'olMail ... 43
39 If (objItem.Class = olMail) And (objItem.Subject Like "*Excel*") Then
40
41 'počítadlo
42 i = i + 1
43
44 'přidání informací do pole
45 'čas doručení
46 arrPoleData(i, 1) = objItem.ReceivedTime
47
48 'jméno
49 arrPoleData(i, 2) = objItem.SenderName
50
51 'e-mail
52 arrPoleData(i, 3) = objItem.SenderEmailAddress
53
54 'předmět zprávy
55 arrPoleData(i, 4) = objItem.Subject
56
57 'obsah
58 arrPoleData(i, 5) = objItem.Body
59
60 'kopie CC, BCC
61 arrPoleData(i, 6) = objItem.CC
62
63 End If
64
65 Next objItem
66
67 'vložení informací do listu
68 wshZpravy.Cells(1).Resize(lngPocetPolozek, 4) = arrPoleData
69
70 'odstranění proměnných z paměti
71 Set objFolder = Nothing
72 Set objNameSpace = Nothing
73 Set objOutlook = Nothing
74
75 End Sub

```

Něco praktičtějšího? Tak projdeme všechny nepřečtené zprávy, roztřídíme je a uložíme přílohy do složky podle odesílatele.

```

1 'Tools / References / Microsoft Outlook xx.x Object Library
2
3 Private Const cstrCesta As String = "D:\Test"
4
5 Sub OutlookUlozeniPrilohDleOdesilatele()
6
7     Dim cstrCestaPriloha As String
8
9     Dim objOutlook As Object
10    Dim objNameSpace As Object
11    Dim objFolderInbox As Object
12    Dim objFolderDeletedItems As Object
13    Dim objItem As Object
14
15    Set objOutlook = New Outlook.Application
16    Set objNameSpace = objOutlook.GetNamespace("MAPI")
17    'olFolderInbox ... 6, olFolderDeletedItems ... 3
18    Set objFolderInbox = objNameSpace.GetDefaultFolder(olFolderInbox)
19    Set objFolderDeletedItems = _
20        objNameSpace.GetDefaultFolder(olFolderDeletedItems)
21
22    'pro každou položku v adresáři
23    For Each objItem In objFolderInbox.Items
24
25        'je zpráva doposud nepřečtená?
26        If objItem.UnRead Then
27
28            With objItem.Attachments
29
30                'existuje příloha?
31                If .Count > 0 Then
32
33                    'pro všechny přílohy
34                    For i = 1 To .Count
35
36                        'cstrCesta ke složce odesílatele
37                        cstrCestaPriloha = cstrCesta & objItem.SenderName & ""
38
39                        'vytvoření (neexistující) složky
40                        On Error Resume Next
41                        MkDir cstrCestaPriloha
42
43                        'uložení přílohy
44                        .Item(i).SaveAsFile (cstrCestaPriloha & .Item(i).Filename)
45
46                    Next i
47
48                End If
49
50            End With
51
52            'nastavení atributu přečtené zprávy
53            objItem.UnRead = False
54
55            'přesun zprávy do složky Odstraněná pošta
56            objItem.Move (objFolderDeletedItems)
57
58            'odstranění zprávy
59            objItem.Delete
60
61        End If
62
63    Next objItem
64
65    'ukončení seance
66    objOutlook.Quit
67
68 End Sub

```


Všechny dosavadní příklady fungovaly ve směru z Outlooku do Excelu. Poslední úloha bude opačná. Ze seznamu úkolů v Excelu vytvoříme úkol/událost v Outlooku. Níže uvedená procedura je jiná ještě v jedné věci. Ukazuje druhý způsob, jak se odkázat na aplikaci Outlook bez nutnosti reference (metoda CreateObject). Zatímco v předchozích ukázkách s referencemi jsme si mohli dovolit používat konstanty Outlooku a deklarovat objekty i přímo a nejen jako obecný Object, zde to možné není.



Propojení Excel - Outlook - událost

```

1  Sub OutlookVytvoritUkolUdalost()
2
3  Dim objOutlook As Object
4  Dim objTaskItem As Object
5  Dim objAppointmentItem As Object
6
7  Dim intRadek As Integer
8
9  'aktivace patřičného listu
10 wshUdalosti.Activate
11
12 'volba datového řádku
13 intRadek = 2
14
15 'vytvoření instance aplikace Outlook
16 Set objOutlook = CreateObject("Outlook.Application")
17
18 'úkol ... olTaskItem ... 3
19 Set objTaskItem = objOutlook.CreateItem(3)
20
21 With objTaskItem
22     .Subject = Cells(intRadek, 1).Text
23     .StartDate = Cells(intRadek, 2).Value
24     .DueDate = Cells(intRadek, 3).Value
25     .ReminderTime = Cells(intRadek, 4).Value
26     .Body = Cells(intRadek, 5).Text
27     .Save
28 End With
29
30 'událost ... olAppointmentItem ...1
31 Set objAppointmentItem = objOutlook.CreateItem(1)
32
33 With objAppointmentItem
34     'předmět
35     .Subject = Cells(intRadek, 1).Text
36     'počáteční datum
37     .Start = Cells(intRadek, 2).Value
38     'konečné datum
39     .End = Cells(intRadek, 3).Value + 1
40     'celodenní událost
41     .AllDayEvent = True
42     'upozornění v minutách před poč. datem
43     .ReminderMinutesBeforeStart = (Cells(intRadek, 2).Value - _
44         Cells(intRadek, 4).Value) * 1440
45     'obsah
46     .Body = Cells(intRadek, 5).Text
47     'místo
48     .Location = Cells(intRadek, 6).Text
49     'uložení
50     .Save
51 End With
52
53 'odstranění objektů z paměti
54 Set objTaskItem = Nothing
55 Set objAppointmentItem = Nothing
56 Set objOutlook = Nothing
57
58 End Sub

```



Propojení Excel - Outlook - událost

Chcete-li si nastudovat objektový model Outlooku, můžete zavítat na stránky [Object model \(Outlook VBA reference\)](#). Pokud vás to navnadilo podívat se na Outlook blíže, pak jen dobře. Uvědomte si, že i Outlook má události a vy můžete po příchodu zprávy automaticky uložit přílohu a mail zpracovat...

Příloha:

[excel_propojeni_outlook.zip](#)